Lagrangian DAE Modelling Examples

Robert Piché
Tampere University of Technology

July 7, 1999

This text presents some examples of physical systems that are modeled using the Lagrangian DAE technique described by Layton [2], and simulated using Dynast [3].

1 Introduction

This section summarises the Lagrangian DAE modeling technique. Layton's book [2] gives further details.

In a Lagrangian DAE model there are $2n+m_1+m_2+m_3+m_4$ state variables, as follows:

- displacement variables $q_{1 \ldots n}$
- flow variables $f_{1 \ldots n}$
- displacement constraint Lagrange multipliers $\kappa_{1 \ldots m_1}$
- flow constraint Lagrange multipliers $\mu_{1 \ldots m_2}$
- implicit efforts ${e^\gamma}_{1 \ldots m_3}$
- dynamic variables $s_{1 \ldots m_4}$

The interpretations of displacement and flow variables in various physical domains are listed in Table 1. Displacement and Flow variables are called kinematic variables.

A system model consists of the following functions:

- potential energy $V(q,t)$
- kinetic coenergy $T^*(f,q,t)$

Domain	Displacement q	Flow $f = \dot{q}$	Effort e
mechanical translation	position	velocity	force
mechanical rotation	angle	angular velocity	torque
electrical	charge	current	voltage
incompressible fluid	fluid volume	volumetric flow rate	pressure

Table 1: Unified notation for variables in various physical domains



Figure 1: Atwood's machine

- content $D(f, q, t)$
- applied efforts $[Q_{1\ldots n}] = Q(e^\gamma, t)$
- displacement constraint functions $[\phi_{1\ldots m_1}] = \Phi(q, t)$
- flow constraint functions $[\psi_{1\ldots m_2}] = \Psi(f, q, t)$
- implicit effort constraint functions $[\gamma_{1\ldots m_3}] = \Gamma(e^\gamma, s, f, q, t)$
- dynamic variable derivative functions $[\lambda_{1\ldots m_4}] = \Lambda(e^\gamma, s, f, q, t)$

The system model gives all the information needed to assemble the Lagrangian DAE.

The Lagrangian DAE is generated from the system model according to the following formula:

$$\begin{aligned}
\dot{q} - f &= f \\
(\nabla_q^2 T^*)\dot{f} - Q + (\nabla_f T^*)_q f + (\nabla_f T^*)_t - \nabla_q T^* & \\
+\nabla_q V + \nabla_f D + \Phi_q^T \kappa + \Psi_f^T \mu &= 0 \\
\Phi &= 0 \\
\Psi &= 0 \\
\Gamma &= 0 \\
\dot{s} - \Lambda &= 0
\end{aligned}$$

This set of DAEs can be entered into Dynast for simulation.

2 Mechanical Examples

2.1 Atwood's Machine

2.1.1 Description

Two bobs are connected by a light rigid cord that passes without slipping over a frictionless pulley (Figure 1). The pulley is restrained by a torsional spring. The following parameters are given:

$$
\begin{aligned}
L &= \text{length of cord} = 1.2 \text{ m} \\
g &= \text{gravitational acceleration} = 9.81 \text{ m s}^{-2} \\
r &= \text{radius of pulley} = 0.2 \text{ m} \\
M_1 &= \text{mass of bob on the left} = 0.1 \text{ kg} \\
M_2 &= \text{mass of bob on the right} = 0.2 \text{ kg} \\
I &= \text{moment of inertia of pulley} = 0.01 \text{ kg m}^2 \\
k &= \text{torsional stiffness of pulley axle} = \text{N m rad}^{-1}
\end{aligned}
$$

The bobs are initially at equal height and are at rest. Find the angle of the pulley as a function of time.

2.1.2 Model

Kinematic variables are labelled and oriented as shown in Figure 1. Variables 1–2 are translational, variable 3 is rotational.

The bobs contribute the following terms to the potential energy and to the kinetic coenergy:

$$
\begin{aligned}
V_{\text{bobs}} &= -M_1 g q_1 - M_2 g q_2 \\
T^*_{\text{bobs}} &= \frac{1}{2} M_1 f_1^2 + \frac{1}{2} M_2 f_2^2
\end{aligned}
$$

The pulley contributes the following terms:

$$
\begin{aligned}
V_{\text{pulley}} &= \frac{1}{2} k q_3^2 \\
T^*_{\text{pulley}} &= \frac{1}{2} I f_3^2
\end{aligned}
$$

The cord consists of three segments: the curved segment lying on the pulley, and the two vertical segments between the pulley and the bobs. The total cord length is

$$
L = (q_1 - r) + (q_2 - r) + \pi r \tag{1}
$$

Differentiating (1) gives a flow constraint

$$
f_1 + f_2 = 0
$$

The nonslipping cord relates the rotational velocity of the pulley to the translational velocity of the bob on the right, according to

$$
r f_3 - f_2 = 0
$$

In summary, there are 3 kinematic variables and 2 flow constraints. The system model is

$$
\begin{aligned}
V &= -M_1 g q_1 - M_2 g q_2 + \frac{1}{2} k q_3^2 \\
T^* &= \frac{1}{2} M_1 f_1^2 + \frac{1}{2} M_2 f_2^2 + \frac{1}{2} I f_3^2 \\
\psi_1 &= f_1 + f_2 \\
\psi_2 &= r f_3 - f_2
\end{aligned}
$$

The initial condition is found by substituting $q_1 = q_2$ into (1) and solving, yielding

$$q_1(0) = q_2(0) = \frac{L - \pi r}{2} + r$$

2.1.3 Simulation

From the model the following equations of motion are derived:

$$\begin{aligned}
\dot{q}_i - f_i &= 0 \ \ (i = 1, 2, 3) \\
M_1 \dot{f}_1 - M_1 g + \mu_1 &= 0 \\
M_2 \dot{f}_2 - M_2 g + \mu_1 - \mu_2 &= 0 \\
I \dot{f}_3 + k q_3 + r \mu_2 &= 0 \\
f_1 + f_2 &= 0 \\
r f_3 - f_2 &= 0
\end{aligned}$$

The DYNAST program is

```
*SYSTEM;

: Parameters
g = 9.81;   :[m/s^2] gravitational acceleration
r = 0.2;    :[m] radius of pulley
L = 1.2;    :[m] string length
M1 = 0.1;   :[kg] mass of bob on the left
M2 = 0.2;   :[kg] mass of bob on the right
I = 0.01;   :[kg m^2] moment of inertia of pulley
k = 1;      :[N m/rad] torsional stiffness of pulley axle

: Equations of motion
SYSVAR q1,q2,q3,f1,f2,f3,mu1,mu2;
0 = VD.q1 - f1;
0 = VD.q2 - f2;
0 = VD.q3 - f3;
0 = M1*VD.f1 - g*M1 + mu1;
0 = M2*VD.f2 - g*M2 + mu1 - mu2;
0 = I*VD.f3 + k*q3 + r*mu2;
0 = f1 + f2;
0 = r*f3 - f2;

: simulation
*TR;
tr 0 3;         : time=0..3 s
INIT q1=.486,q2=.486;
PRINT(100) q1,q2;
RUN;

*END;
```

The simulation results (Fig. 2) show the motion of the bobs. They appear to oscillate with a period of approximately 0.9 s.



Figure 2: Simulation results for Atwood's machine



Figure 3: Ball on wheel

2.1.4 Reference

This example is from [1], with the addition of torsional stiffness to the pulley, numerical values for the parameters, and initial conditions.

2.1.5 Date

July 7, 1999

2.2 Ball on Wheel

2.2.1 Description

A cylinder (wheel) rotates about its longitudinal axis, and a ball is rolling without slipping on the surface of the cylinder (Figure 3). The ball is initially on top of the wheel. A torque is applied to the wheel. The following parameters and functions are given:

$$\begin{aligned}
g &= \text{gravitational acceleration} = 9.81 \text{ m s}^{-2} \\
r_{\mathrm{w}} &= \text{radius of wheel} = 1 \text{ m} \\
r_{\mathrm{b}} &= \text{radius of ball} = 0.2 \text{ m} \\
m_{\mathrm{b}} &= \text{mass of ball} = 1 \text{ kg} \\
J_{\mathrm{b}} &= \text{moment of inertia of ball} = \frac{2}{5} m_{\mathrm{b}} r_b^2
\end{aligned}$$

$$
\begin{aligned}
J_{\mathrm{w}} &= \text{moment of inertia of wheel} = 0.1\ \mathrm{kg\ m}^2 \\
M(t) &= \text{torque applied to wheel} = 0.2\sin((20\ \mathrm{s}^{-1})t)\ \mathrm{N\ m}
\end{aligned}
$$

Find the angles as functions of time when the ball starts from rest at the top of the wheel.

2.2.2 Model

Kinematic variables are labelled and oriented as shown in Figure 1. All 3 variables are rotational.

The ball's gravitational potential energy is $V = m_{\mathrm{b}}gh$ where h is the height of the ball's centre of mass, given by

$$h = (r_{\mathrm{b}} + r_{\mathrm{w}})\cos q_3$$

The kinetic coenergy of the ball due to its translation is

$$T^*_{\mathrm{trans}} = \frac{1}{2}m_{\mathrm{b}}v^2$$

where the speed of the ball is

$$v = (r_{\mathrm{b}} + r_{\mathrm{w}})f_3$$

The kinetic coenergy due to rotation of the wheel and rotation of the ball is

$$T^*_{\mathrm{rot}} = \frac{1}{2}J_{\mathrm{w}}f_1^2 + \frac{1}{2}J_{\mathrm{b}}f_2^2$$

Because the ball rolls without slipping, we have the condition

$$r_{\mathrm{w}}q_1 + r_{\mathrm{b}}q_2 = (r_{\mathrm{b}} + r_{\mathrm{w}})q_3$$

Differentiating this and rearranging gives the flow constraint

$$r_{\mathrm{w}}f_1 + r_{\mathrm{b}}f_2 - (r_{\mathrm{b}} + r_{\mathrm{w}})f_3 = 0$$

The torque applied to the wheel is a source effort corresponding to the wheel's rotation:

$$Q_1 = M(t)$$

In summary, there are 3 kinematic variables and there is 1 flow constraint. The system model is

$$
\begin{aligned}
V &= m_{\mathrm{b}}g(r_{\mathrm{b}} + r_{\mathrm{w}})\cos q_3 \\
T^* &= \frac{1}{2}J_{\mathrm{w}}f_1^2 + \frac{1}{2}J_{\mathrm{b}}f_2^2 + \frac{1}{2}m_{\mathrm{b}}(r_{\mathrm{b}} + r_{\mathrm{w}})^2 f_3^2 \\
Q &= [M(t), 0, 0]^T \\
\psi_1 &= r_{\mathrm{w}}f_1 + r_{\mathrm{b}}f_2 - (r_{\mathrm{b}} + r_{\mathrm{w}})f_3
\end{aligned}
$$

2.2.3 Simulation

From the model the following equations of motion are derived:

$$
\begin{aligned}
\dot{q}_i - f_i &= 0 \quad (i = 1, 2, 3) \\
J_{\mathrm{w}} \dot{f}_1 + r_{\mathrm{w}} \mu_1 - M(t) &= 0 \\
J_{\mathrm{b}} \dot{f}_2 + r_{\mathrm{b}} \mu_1 &= 0 \\
m_{\mathrm{b}} (r_{\mathrm{b}} + r_{\mathrm{w}})^2 \dot{f}_3 - m_{\mathrm{b}} g (r_{\mathrm{b}} + r_{\mathrm{w}}) \sin q_3 - (r_{\mathrm{b}} + r_{\mathrm{w}}) \mu_1 &= 0 \\
r_{\mathrm{w}} f_1 + r_{\mathrm{b}} f_2 - (r_{\mathrm{b}} + r_{\mathrm{w}}) f_3 &= 0
\end{aligned}
$$

The DYNAST program is

```
*SYSTEM;

: Parameters and functions
g = 9.81;           :[m/s^2] gravitational acceleration
rw = 1.0;           :[m] radius of wheel
rb = 0.2;           :[m] radius of ball
mb = 1.0;           :[kg] mass of ball
Jb = 0.4*mb*rb**2;  :[kg.m^2] moment of inertia of ball
Jw = 0.1;           :[kg.m^2] moment of inertia of wheel
M/sin/B=0.2,C=20;   :[N.m] torque applied to wheel

: Equations of motion
SYSVAR q1,q2,q3,f1,f2,f3,mu1;
0 = VD.q1 - f1;
0 = VD.q2 - f2;
0 = VD.q3 - f3;
0 = Jw*VD.f1 + rw*mu1- M(TIME);
0 = Jb*VD.f2 + rb*mu1;
0 = mb*(rb+rw)**2*VD.f3 - (rb+rw)*(mb*g*sin(q3)+mu1);
0 = rw*f1 + rb*f2 - (rb+rw)*f3;

: simulation
*TR;
tr 0 1;         : time=0..1 s
PRINT(100) q1,q2,q3;
RUN;

*END;
```

The simulation results (Fig. 4) show how the ball starts to move. The position at the top of the wheel appears to be unstable.

2.2.4 Reference

This example is from [1], with the addition of numerical values for the parameters.

2.2.5 Date

July 7, 1999



Figure 4: Simulation results for ball on wheel



Figure 5: A simple RLC circuit.

3 Electric Circuit Examples

3.1 RLC Circuit

3.1.1 Description

A simple RLC circuit is shown in Figure 5. The following parameters are given.

$$\begin{aligned} C_1 &= \text{capacitance for } q_1 = 2\ \mu\text{F} \\ C_2 &= \text{capacitance for } q_2 = 1\ \mu\text{F} \\ R &= \text{resistance} = 10\ \Omega \\ L &= \text{inductance} = 100\ \text{H} \\ u_0(t) &= \text{voltage source} = \begin{cases} 10\ \text{V} & 0 < t \leq 20\ \text{ms} \\ 0 & 20\ \text{ms} < t \end{cases} \end{aligned}$$

Find the currents in the circuit as functions of time.

3.1.2 Model

Kinematic variables are labelled and oriented as shown in Figure 5. All variables are electrical.

The capacitors contribute terms to the potential energy

$$V = \frac{1}{2}\frac{1}{C_1}q_1^2 + \frac{1}{2}\frac{1}{C_2}q_2^2$$

the resistor gives a content function

$$D = \frac{1}{2}Rf_4^2$$

and the coil gives a kinetic coenergy.

$$T^* = \frac{1}{2} L f_3^2$$

The voltage source is an applied effort corresponding to displacement q_5, so that

$$Q_5 = u_0(t)$$

At the nodes, the sum of incoming displacements and flows equals the sum of outgoing displacements and flows. Because three of the nodes are connected to capacitances, they are modeled as displacement constraints:

$$\begin{aligned} -(q_1 - q_1(0)) - q_5 &= 0 \\ (q_1 - q_1(0)) - (q_2 - q_2(0)) - q_3 &= 0 \\ (q_2 - q_2(0)) - q_4 &= 0 \end{aligned}$$

The grounded node is not connected to a capacitance, so it is modelled as a flow constraint:

$$f_3 + f_4 + f_5 - f_6 = 0$$

In summary, there are 6 kinematic variables, 3 displacement constraints, and 1 flow constraint. The system model is

$$\begin{aligned} V &= \frac{1}{2}\frac{1}{C_1} q_1^2 + \frac{1}{2}\frac{1}{C_2} q_2^2 \\ D &= \frac{1}{2} R f_4^2 \\ T^* &= \frac{1}{2} L f_3^2 \\ Q &= [0, 0, 0, 0, u_0(t), 0]^T \\ \phi_1 &= -(q_1 - q_1(0)) - q_5 \\ \phi_2 &= (q_1 - q_1(0)) - (q_2 - q_2(0)) - q_3 \\ \phi_3 &= (q_2 - q_2(0)) - q_4 \\ \psi_1 &= f_3 + f_4 + f_5 - f_6 \end{aligned}$$

3.1.3 Simulation

From the model the following equations of motion are derived:

$$\begin{aligned} \dot{q}_i - f_i &= 0 \;\; (i = 1, \ldots, 6) \\ \frac{1}{C_1} q_1 - \kappa_1 + \kappa_2 &= 0 \\ \frac{1}{C_2} q_2 - \kappa_2 - \kappa_3 &= 0 \\ L\dot{f}_3 - \kappa_2 + \mu_1 &= 0 \\ R f_4 - \kappa_3 + \mu_1 &= 0 \\ -\kappa_1 + \mu_1 - u_0(t) &= 0 \\ -(q_1 - q_1(0)) - q_5 &= 0 \\ (q_1 - q_1(0)) - (q_2 - q_2(0)) - q_3 &= 0 \\ (q_2 - q_2(0)) - q_4 &= 0 \\ f_3 + f_4 + f_5 - f_6 &= 0 \end{aligned}$$

The DYNAST program is

```
*SYSTEM;

: Parameters and functions
C1=2u;     :[F] capacitance for q1
C2=1u;     :[F] capacitance for q2
R=10K;     :[ohm] resistance
L= 100;    :[H] inductance
U0 /pulse/ L2=10, TT = 20u; : [V] voltage source
q1ini = 0; :[coul] initial value of charge q1
q2ini = 0; :[coul] initial value of charge q2

: Equations of motion
SYSVAR q1,q2,q3,q4,q5,q6,
       f1,f2,f3,f4,f5,f6,
       kappa1,kappa2,kappa3,mu1;
0 = VD.q1 - f1;
0 = VD.q2 - f2;
0 = VD.q3 - f3;
0 = VD.q4 - f4;
0 = VD.q5 - f5;
0 = VD.q6 - f6;
0 = q1/C1 - kappa1 + kappa2;
0 = q2/C2 - kappa2 + kappa3;
0 = L*VD.f3 - kappa2 + mu1;
0 = R*f4 - kappa3 + mu1;
0 = -kappa1 + mu1 - u0(TIME);
0 = -mu1;
0 = -(q1-q1ini) - q5;
0 = (q1-q1ini) - (q2-q2ini) - q3;
0 = (q2-q2ini) - q4;
0 = f3 + f4 + f5 - f6;

: simulation
*TR;
tr 0 1;         : time=0..1 s
PRINT(100) f1,f2,f3;
RUN;

*END;
```

The simulation results (Fig. 6) show that the currents are all damped oscillations.

3.1.4 Reference

This is Worked Out Example 11 from [1], with the addition of a ground and of numerical values.



Figure 6: Simulation results for RLC circuit



Figure 7: Capacitor with moving plate.

3.1.5 Date

July 7, 1999

4 Electromechanical Examples

4.1 Capacitor with moving plate

4.1.1 Description

A simple electric circuit consists of a voltage source, a resistor, and a capacitor in series (Figure 7). In the capacitor, one plate is fixed, but the other plate moves, and it is held in place with a mechanical spring. The capacitance is inversely proportional to the plate separation distance. The given parameters are

$$\begin{aligned}
c &= \text{the product of capacitance and separation distance} = 2\ \mu\text{F m} \\
k &= \text{stiffness of spring} = 1\ \text{kN m}^{-1} \\
R &= \text{resistance} = 10\ \text{k}\Omega \\
m &= \text{mass of moving plate} = 1\ \text{g} \\
u_0(t) &= \text{voltage source} = \begin{cases} 10\ \text{V} & 0 < t \leq 20\ \text{ms} \\ 0 & 20\ \text{ms} < t \end{cases}
\end{aligned}$$

Find the motion of the moving plate as a function of time.

4.1.2 Model

Kinematic variables are labelled and oriented as shown in Figure 7. Variables 1–4 are electrical (charge, current), variables 5–6 are translational mechanical (position, velocity).

The capacitor and spring contribute terms to the potential energy

$$V = \frac{1}{2}\frac{1}{C}q_3^2 + \frac{1}{2}kq_6^2$$

The capacitance is inversely proportional to the separation distance, so

$$C = \frac{c}{q_5}$$

The resistor gives a content

$$D = \frac{1}{2} R f_2^2$$

and the plate mass gives a kinetic coenergy

$$T^* = \frac{1}{2} m f_5^2$$

The voltage source is an applied effort corresponding to displacement q_1, so that

$$Q_1 = u_0(t)$$

At the nodes, the sum of incoming displacements and flows equals the sum of outgoing displacements and flows. Because three nodes are connected to capacitances, they are modeled as displacement constraints:

$$\begin{aligned} q_2 - (q_3 - q_3(0)) &= 0 \\ q_1 + (q_3 - q_3(0)) &= 0 \\ q_5 - (q_6 - q_6(0)) &= 0 \end{aligned}$$

The grounded node is not connected to a capacitance, so it is modelled as a flow constraint:

$$-f_1 - f_2 - f4 = 0$$

In summary, there are 6 kinematic variables, 3 displacement constraints, and 1 flow constraint. The system model is

$$\begin{aligned} V &= \frac{1}{2} \frac{q_5}{c} q_3^2 + \frac{1}{2} k q_6^2 \\ D &= \frac{1}{2} R f_2^2 \\ T^* &= \frac{1}{2} m f_5^2 \\ Q &= [u_0(t), 0, 0, 0, 0, 0]^T \\ \phi_1 &= q_2 - (q_3 - q_3(0)) \\ \phi_2 &= q_1 + (q_3 - q_3(0)) \\ \phi_3 &= q_5 - (q_6 - q_6(0)) \\ \psi_1 &= -f_1 - f_2 - f4 \end{aligned}$$

4.1.3 Simulation

From the model the following equations of motion are derived:

$$\begin{aligned} \dot{q}_i - f_i &= 0 \quad (i = 1, \ldots, 6) \\ \kappa_2 - \mu_1 - u_0(t) &= 0 \\ R f_2 + \kappa_1 - \mu_1 &= 0 \end{aligned}$$

$$\begin{aligned}
\frac{q_5}{c}q_3 - \kappa_1 + \kappa_2 &= 0 \\
-\mu_1 &= 0 \\
m\dot{f}_5 + \frac{1}{2}\frac{1}{c}q_3^2 + \kappa_3 &= 0 \\
kq_6 - \kappa_3 &= 0 \\
q_2 - (q_3 - q_3(0)) &= 0 \\
q_1 + (q_3 - q_3(0)) &= 0 \\
q_5 - (q_6 - q_6(0)) &= 0 \\
-f_1 - f_2 - f4 &= 0
\end{aligned}$$

The DYNAST program is

```
*SYSTEM;

: Parameters and functions
c=2u;       :[F.m] constant, capacitance * distance
k=1K;       :[N/m] stiffness
R=10K;      :[ohm] resistance
m= 1e-3;    :[kg] mass
U0 /pulse/ L2=10, TT = 20u; :[V] voltage source
q3ini = 0; :[coul] initial value of charge q3
q6ini = 0; :[m] initial value of displacement q6

: Equations of motion
SYSVAR q1,q2,q3,q4,q5,q6,
       f1,f2,f3,f4,f5,f6,
       kappa1,kappa2,kappa3,mu1;
0 = VD.q1 - f1;
0 = VD.q2 - f2;
0 = VD.q3 - f3;
0 = VD.q4 - f4;
0 = VD.q5 - f5;
0 = VD.q6 - f6;
0 = kappa2 - mu1 - U0(TIME);
0 = R*f2 + kappa1 - mu1;
0 = q5*q3/c - kappa1 + kappa2;
0 = -mu1;
0 = m*VD.f5 + 0.5*q3**2/c + kappa3;
0 = k*q6 - kappa3;
0 = q2 - (q3-q3ini);
0 = q1 + (q3-q3ini);
0 = q5 - (q6-q6ini);
0 = -f1 - f2 - f4;

: simulation
*TR;
tr 0 20m;           : time=0..20 ms
PRINT(100) q5;
RUN;
```



Figure 8: Simulation results for capacitor with moving plate



Figure 9: Fluid accumulator discharge example

```
*END;
```

The simulation results (Fig. 8) show that plate motion is an undamped oscillation.

4.1.4 Reference

This example is from [1], with the addition of a ground and of numerical values.

4.1.5 Date

July 7, 1999

5 Fluid Circuit Examples

5.1 Fluid accumulator discharge

5.1.1 Description

A hydraulic accumulator (a capacitance element) with a single orifice vents compressible fluid to a reservoir with zero pressure (Figure 9). The orifice flow is turbulent, that is, flow rate is proportional to the square root of the pressure difference. The following parameters are given:

$$\begin{aligned} \beta &= \text{bulk modulus} = 1 \text{ GPa} \\ \text{Vol} &= \text{volume of accumulator} = 0.01 \text{ m}^3 \\ C_d &= \text{orifice discharge coefficient} = 0.6 \end{aligned}$$

$$
\begin{aligned}
A &= \text{orifice area} = 50 \text{ mm}^2 \\
\rho &= \text{fluid mass density} = 900 \text{ kg } m^{-3} \\
P(0) &= \text{initial accumulator pressure} = 1 \text{ MPa}
\end{aligned}
$$

Find the accumulator pressure as a function of time.

5.1.2 Model

Kinematic variables are labelled and oriented as shown in Figure 9.

The accumulator gives rise to a potential energy

$$V = \frac{1}{2}\frac{\beta}{\text{Vol}}q_1^2$$

while the orifice dissipates energy according to a nonideal content function

$$D = \frac{1}{3}\frac{\rho}{2C_d^2A^2}|f_2|^3$$

At the node connecting the orifice to the accumulator, the constraint equation is

$$-(q_1 - q_1(0)) - q_2 = 0$$

In summary, the system model has 2 kinematic variables and 1 displacement constraint. The system model is

$$
\begin{aligned}
V &= \frac{1}{2}\frac{\beta}{\text{Vol}}q_1^2 \\
D &= \frac{1}{3}\frac{\rho}{2C_d^2A^2}|f_2|^3 \\
\phi &= -(q_1 - q_1(0)) - q_2
\end{aligned}
$$

The accumulator pressure is a function of the fluid quantity,

$$P = \frac{\beta}{\text{Vol}}q_1$$

and this can be solved to give the initial displacement

$$q_1(0) = \frac{\text{Vol}}{\beta}P(0)$$

5.1.3 Simulation

From the model the following equations of motion are derived:

$$
\begin{aligned}
\dot{q}_i - f_i &= 0 \quad (i = 1, 2) \\
\frac{\beta}{\text{Vol}}q_1 - \kappa_1 &= 0 \\
\frac{\rho}{2C_d^2A^2}|f_2|f_2 - \kappa_1 &= 0 \\
-(q_1 - q_1(0)) - q_2 &= 0
\end{aligned}
$$

The DYNAST program is



Figure 10: Simulation results for accumulator discharge example.

```
*SYSTEM;

: Parameters and functions
beta = 1G;     :[Pa] bulk modulus
Vol = 0.01;    :[m^3] volume of accumulator
Cd = 0.6;      :[1] orifice discharge coefficient
A = 50u;       :[m^2] orifice area
rho = 900;     :[kg/m^3] fluid mass density
q1ini = 10u;   :[m^3] initial amount of fluid in capacitor

: Equations of motion
SYSVAR q1,q2,f1,f2,kappa1;
0 = VD.q1 - f1;
0 = VD.q2 - f2;
0 = beta/Vol*q1 - kappa1;
0 = rho/(2*Cd**2*A**2)*abs(f2)*f2 - kappa1;
0 = -(q1-q1ini) - q2;

: Derived quantities
P = beta/Vol*q1;    : pressure in accumulator

: simulation
*TR;
tr 0 20m;          : time=0..20 ms
INIT q1=10u;
PRINT(100) P;
RUN;

*END;
```

The simulation results (Fig. 10) show that the pressure drops to zero in approximately 0.014s.

5.1.4 Reference

This example is from [4], where a more complicated model of the orifice flow is needed to avoid numerical difficulties with ODE solvers when the pressure is zero.

Note that Layton[2, p.135] models nonlinear fluid resistance as an internal effort constraint; here it is modeled using a content function.

Layton's models of accumulators are for incompressible fluids and so cannot be used here. The model for compressible fluids is however of the same form.



Figure 11: Hydraulic cylinder test bench

5.1.5 Date

July 7, 1999

6 Fluid-Mechanical Examples

6.1 Hydraulic cylinder test bench

6.1.1 Description

In the system illustrated in Figure 11, a hydraulic cylinder is subjected to a varying fluid flow that is fed through a valve that is gradually opening. The flow through the valve is turbulent; the flow through the restrictions is laminar. The cylinder's movement is resisted by a spring. Given quantities are

$$
\begin{aligned}
\beta &= \text{fluid bulk modulus} = 700\ \text{MPa} \\
A_c &= \text{piston cross sectional area} = 0.01\ \text{m}^2 \\
L &= \text{piston offset} = 2.5\ \text{m} \\
m &= \text{piston mass} = 0 \\
k &= \text{spring stiffness} = 500\ \text{kN m}^{-1} \\
C_d &= \text{valve discharge coefficient} = 0.61 \\
A(t) &= \text{valve orifice area} = (10^{-6} + (0.03\ \text{s}^{-1})t)\ \text{m}^2 \\
\rho &= \text{fluid mass density} = 800\ \text{kg m}^{-3} \\
R_1 &= \text{linear fluid resistance for flow } f_1 = 1/(3 \cdot 10^{-8}\ \text{m}^3\ \text{s}^{-1}\ \text{Pa}^{-1}) \\
R_3 &= \text{linear fluid resistance for flow } f_3 = 1/(2 \cdot 10^{-7}\ \text{m}^3\ \text{s}^{-1}\ \text{Pa}^{-1}) \\
f^s(t) &= \text{source fluid flow} = \begin{cases} 0.125\ \text{m}^3\ \text{s}^{-1} & 0 < t \le 0.04\ \text{s} \\ 0 & 0.04\ \text{s} < t \le 0.05\ \text{s} \\ 0.125\ \text{m}^3\ \text{s}^{-1} & 0.05\ \text{s} < t \end{cases}
\end{aligned}
$$

Find the flow rates as functions of time.

6.1.2 Model

Kinematic variables are labelled and oriented as shown in Figure 11.

The mass of the cylinder piston gives a kinetic coenergy

$$T^* = \frac{1}{2} m f_5^2$$

The cylinder pressure $\mathcal{P}$, volume Vol, and flow rate $\mathcal{Q}$ are related by the equation

$$\frac{d\mathcal{P}}{dt} = \frac{\mathcal{Q} - d\text{Vol}/dt}{\text{Vol}} \beta \tag{2}$$

The volume is given by

$$\text{Vol} = A_\text{c}(L + x)$$

where L is the piston offset and A_c is the piston area.

In uniform system notation with $q_4 := \mathcal{V}$, $q_5 := x$, and $s := \mathcal{P}/\beta$, the cylinder equation (2) is a dynamic variable differential equation

$$\dot{s} = \frac{f_4 - A_\text{c} f_5}{A_\text{c}(L + q_5)}$$

Identifying the intrinsic effort $e_1^\gamma := -\mathcal{P}$, the intrinsic effort constraint

$$\beta s + e_1^\gamma = 0$$

follows from the definition of s.

The relation between the force e_2^γ applied to the piston and the pressure in the cylinder is

$$e_2^\gamma + A_\text{c} e_1^\gamma = 0$$

and this is the second intrinsic effort constraint.

The valve dissipates energy according to a nonideal content function

$$D_\text{valve} = \frac{1}{3} \frac{\rho}{2 C_d^2 A^2} |f_2|^3$$

The linear resistance elements dissipate energy according to ideal content functions:

$$D_\text{linear} = \frac{1}{2} R_1 f_1^2 + \frac{1}{2} R_3 f_3^2$$

Two of the nodes are connected to capacitance elements, so they are modelled as displacement constraints:

$$\begin{aligned} q_3 - (q_4 - q_4(0)) &= 0 \\ q_5 - (q_6 - q_6(0)) &= 0 \end{aligned}$$

The two other nodes can be modelled as flow constraints, with one of the nodes having a flow source:

$$\begin{aligned} f^s - f_1 - f_2 &= 0 \\ f_2 - f_3 &= 0 \end{aligned}$$

To summarize, the system model has 6 kinematic variables, 2 displacement constraints, 2 flow constraints, 2 implicit effort constraints, and 1 dynamic variable. The model is

$$
\begin{aligned}
V &= \frac{1}{2}kq_6^2 \\
D &= \frac{1}{3}\frac{\rho}{2C_d^2A^2}|f_2|^3 + \frac{1}{2}R_1f_1^2 + \frac{1}{2}R_3f_3^2 \\
T^* &= \frac{1}{2}mf_5^2 \\
Q &= [0,0,0,e_1^\gamma,e_2^\gamma,0]^T \\
\phi_1 &= q_3 - (q_4 - q_4(0)) \\
\phi_2 &= q_5 - (q_6 - q_6(0)) \\
\psi_1 &= f^s - f_1 - f_2 \\
\psi_2 &= f_2 - f_3 \\
\gamma_1 &= \beta s + e_1^\gamma \\
\gamma_2 &= e_2^\gamma + A_\mathrm{c}e_1^\gamma \\
\lambda &= \frac{f_4 - A_\mathrm{c}f_5}{A_\mathrm{c}(L+q_5)}
\end{aligned}
$$

6.1.3 Simulation

From the model the following equations of motion are derived:

$$
\begin{aligned}
\dot{q}_i - f_i &= 0 \quad (i = 1\ldots,6) \\
R_1f_1 - \mu_1 &= 0 \\
\frac{\rho}{2C_d^2A^2}|f_2|f_2 - \mu_1 + \mu_2 &= 0 \\
R_3f_3 - \mu_2 + \kappa_1 &= 0 \\
-\kappa_1 - e_1^\gamma &= 0 \\
m\dot{f}_5 + \kappa_2 - e_2^\gamma &= 0 \\
kq_6 - \kappa_2 &= 0 \\
q_3 - (q_4 - q_4(0)) &= 0 \\
q_5 - (q_6 - q_6(0)) &= 0 \\
f^s - f_1 - f_2 &= 0 \\
f_2 - f_3 &= 0 \\
\beta s + e_1^\gamma &= 0 \\
e_2^\gamma + A_\mathrm{c}e_1^\gamma &= 0 \\
\dot{s} - \frac{f_4 - A_\mathrm{c}f_5}{A_\mathrm{c}(L+q_5)} &= 0
\end{aligned}
$$

The DYNAST program is

```
*SYSTEM;

: Parameters and functions
beta = 700Me;  :[Pa] fluid bulk modulus
```

```
Ac = 0.01;     :[m^2] piston cross sectional area
L = 2.5;       :[m] piston offset
m = 0;         :[kg] piston mass
k = 500K;      :[N/m] spring stiffness
Cd = 0.61;     :[1] valve discharge coefficient
A /poly/ 1u,0.03; :[m^2/s] valve orifice area
rho = 800;     :[kg/m^3] fluid mass density
R1 = 1/(3e-8); :[Pa.s/m^3] fluid resistance for f1
R3 = 1/(2e-7); :[Pa.s/m^3] fluid resistance for f3
fs /tab/ 0,.125,.04,.125,.04,0,.05,0,
     .05,.125; :[m^3/s] source fluid flow
q4ini = 0;     :[m^3] initial amount of fluid in cylinder
q6ini = 0;     :[m] initial deflection of spring

: Equations of motion
SYSVAR q1,q2,q3,q4,q5,q6,
       f1,f2,f3,f4,f5,f6,
       kappa1,kappa2,mu1,mu2,
       s,egamma1,egamma2;
0 = VD.q1 - f1;
0 = VD.q2 - f2;
0 = VD.q3 - f3;
0 = VD.q4 - f4;
0 = VD.q5 - f5;
0 = VD.q6 - f6;
0 = R1*f1 - mu1;
0 = rho/(2*Cd**2*A(TIME)**2)*abs(f2)*f2 - mu1 + mu2;
0 = R3*f3 - mu2 + kappa1;
0 = -kappa1 - egamma1;
0 = m*VD.f5 + kappa2 - egamma2;
0 = k*q6 - kappa2;
0 = q3 - (q4-q4ini);
0 = q5 - (q6-q6ini);
0 = fs(TIME) - f1 - f2;
0 = f2 - f3;
0 = beta*s + egamma1;
0 = egamma2 + Ac*egamma1;
0 = VD.s - (f4-Ac*f5)/(L+q5)/Ac;

: simulation
*TR;
tr 0 0.1;         : time=0..0.1 s
PRINT(100) f1,f2;
RUN;

*END;
```

The simulation results (Fig. 12) show how the flow rates evolve with time.



Figure 12: Simulation results for hydraulic cylinder test bench example.

6.1.4 Reference

This example is taken from the Simulink Automotive Examples booklet [5]. The value of valve area A given in the booklet does not agree with the value in the simulink model on the diskette; the latter was used here because it gives results that agree with the figures in the booklet.

Note that the hydraulic cylinder modeled in Layton [2, p.30] is for incompressible fluid, and so cannot be used here.

For an analysis of this system using multipoles, see

`http://icosym-nt.cvut.cz/courses/robert.html`

6.1.5 Date

July 7, 1999

References

[1] Werner Haas, Kurt Schlacher, Reinhard Gahleitner, *Modeling of Electromechanical Systems*, RichODL unit, 1999.

[2] Richard A. Layton, *Principles of Analytical System Dynamics*, Springer, 1998.

[3] Herman Mann, Dynast User's Manual, 1998.

[4] Robert Piché and Asko Ellman, A modified orifice flow formula for numerical simulation of fluid power systems, ASME Fluid Power and Technology Collected Papers, Volume 3, 1996, pp. 59-64.

[5] The Mathworks Inc., SIMULINK Automotive Examples, 1996.